

# LogoLISS - A Toy Language

LOGO Language of Integers, Sequences and Sets

Daniela da Cruz      Pedro Rangel Henriques

2008, May

## 1 Grammar

```

/***** Program

Liss          --> "PROGRAM" identifier "{" Body  "}"

Body          --> "DECLARATIONS" Declarations
               "STATEMENTS"  Statements

/***** Declarations

Declarations  --> Declaration
               | Declarations Declaration

Declaration   --> Variable_Declaration

/***** Declarations: Variables

Variable_Declaration --> Vars "->" Type ";"

Vars          --> Var
               | Vars "," Var

Var           --> identifier Value_Var

Value_Var     -->
               | "=" Inic_Var

Type          --> "INTEGER"
               | "BOOLEAN"
               | "SET"
               | "ARRAY" "SIZE" number

Inic_Var      --> Constant
               | Array_Definition
               | Set_Definition

Constant     --> Sign number

```

```

        | string
        | "TRUE"
        | "FALSE"
Sign      -->
        | "+"
        | "-"

/***** Declarations: Variables: Array_Definition

Array_Definition  --> "[" Array_Initialization "]"

Array_Initialization --> Elem
                    | Array_Initialization "," Elem

Elem              --> Sign number

/***** Declarations: Variables: Set Definition

Set_Definition    --> "{"
                    Set_Initialization
                    "}"

Set_Initialization -->
                    | identifier "|" Expression

/***** Statements

Statements        --> Statement
                    | Statements Statement

Statement         --> Turtle_Commands
                    | Assignment
                    | Conditional_Statement
                    | Iterative_Statement

/***** Turtle Statement

Turtle_Commands  --> Step
                    | Rotate
                    | Mode
                    | Dialogue
                    | Location

Step              --> "FORWARD" Expression
                    | "BACKWARD" Expression

Rotate           --> "RRIGHT"
                    | "RLEFT"

Mode             --> "PEN" "UP"

```

```

        | "PEN" "DOWN"

Dialogue      --> Say_Statement
              | Ask_Statement

Location     --> "GOTO" number "," number
              | "WHERE" "?"

/***** Assignment Statement *****/

Assignment   --> Variable "=" Expression

Variable     --> identifier Array_Acess

Array_Acess  -->
              | "[" Expression "]"

/***** Expression *****/

Expression   --> Single_Expression
              | Expression Rel_Oper Single_Expression

/***** Single_Expression *****/

Single_Expression --> Term
                  | Single_Expression Add_Op Term

/***** Term *****/

Term         --> Factor
              | Term Mul_Op Factor

/***** Factor *****/

Factor       --> Constant
              | Variable
              | Set_Definition
              | Member
              | SuccOrPred
              | "!" Expression
              | "+" Expression
              | "-" Expression
              | "(" Expression ")"

/***** Operators *****/

Add_Op       --> "+"
              | "-"
              | "||"
              | "++"

Mul_Op       --> "*"

```

```

| "/"
| "&&"
| "***"

Rel_Orp      --> "=="
| "!="
| "<"
| ">"
| "<="
| ">="
| "in"

/***** SuccOrPredd

SuccOrPred   --> SuccPred identifier

SuccPred     --> "SUCC"
| "PRED"

/***** SetOper

Member       --> "ISMEMBER" "(" Expression "," Variable ")"

/***** IO Statements

Say_Statement --> "SAY" "(" Expression ")"

Ask_Statement --> "ASK" "(" string "," Variable ")"

/***** Conditional & Iterative Statements

Conditional_Statement --> IfThenElse_Stat

Iterative_Statement  --> While_Stat

/***** IfThenElse_Stat

IfThenElse_Stat     --> "IF" Expression
                    "THEN" "{" Statements "}"
                    Else_Expression

Else_Expression     -->
                    | "ELSE" "{" Statements "}"

/***** While_Stat

While_Stat          --> "WHILE" "(" Expression ")" "{" Statements "}"

```